



# ARCADIAN

WELCOME TO THE FASCINATING WORLD OF ARCADE PROGRAMMING!

The enclosed programs were selected to help you enjoy your Arcade-Plus game unit, and were originally published in the ARCADIAN. If you review the programs included here against their listing in the ARCADIAN, you will find that we have made some procedural changes in order to be compatible with your new Basic cartridge. These programs are presented as "listings" - all the instructions necessary for the computer to operate, in a standardized format that the computer understands.

These programs have all been checked for operation and you should have no problem with them. Read the Manual that came with the Basic in case you have difficulty - and here are a couple of Instructions that are useful for any program you wish to enter.

1. Make NO modifications.

2. It is implied that every line ends with a 'WORDS GO' entry, that is equivalent to a typewriter's carriage return. Up to this point, you could make changes, correct errors, etc., because the work you were doing was being temporarily stored in a "keypad buffer". Once you press the 'WORDS GO' sequence, that work is transferred to the computer and the keyboard buffer is ready for another line of input.

3. Punctuation marks have a different meaning to the computer. Do not substitute, add, or delete any.

4. After the last line entry, type in "WORDS RUN GO" to get the program into operation.

5. If the computer has a problem and questions your instruction, the instruction will be listed with a question mark at the problem area. First 'LIST' the line in question (for example, if the computer prints line 140 on the screen, enter "LIST 140, 1 WORDS GO". This will list only line 140, just as it was entered.) and see that it is exactly as you wanted it, especially in the area where the question mark was. Next, make sure that the computer can do what you want it to - is it supposed to go to a subroutine that isn't there? or use a variable that has not been entered? are there enough terms for a LINE or BOX command? etc.

6. If the problem is unsolvable, list the program to a tape, send it to us, and we'll return a "fixed" program to you.

7. Once the program is running, you can store it on tape for your future use. And after it is stored, you can start to manipulate the program in the computer and make whatever modifications you might wish.

All programs are available on a single tape, at \$7.50 postpaid.

ARCADIAN Newsletter

**PROGRAM NAME** BAGELS

**ISSUE** Volume 2 page 25

**AUTHOR**

Carl Morimoto  
4003 Hamilton Park Dr.  
San Jose, CA 95130

```
>371 BAGELS
>214 PICO
>124 FERMI
>526 FERMI
>□
```

**DESCRIPTION**

The computer picks a three-digit number (no duplicate digits) that you have to guess. Each time that you make a three-figure guess, the computer responds with one of the following:

BAGELS no digit is correct  
PICO one correct digit in the wrong place  
PICO-PICO two or more correct digits in the wrong place  
FERMI one correct digit in the right place  
FERMI-PICO one correct digit in the right place, and one or more correct digits in the wrong place  
FERMI-FERMI two or more correct digits in the right place

When you think that you have the right numbers, press GO, and the computer will respond. If you want to give up, enter three zeros 000

**LISTING**

```
50 CLEAR ;NT=0;FC=66;BC=7
60 Z=300
70 Q=0;W=0;D=0;C=0;U=0;F=0
100 BOX 5,-5,80,50,1;BOX 5,-5,78,48,3
110 CX=-40;CY=20;PRINT "BAGELS"
120 BOX -26,20,49,11,3
130 NT=5;FOR A=1TO 15
140 CX=RND(3)*20-40;CY=RND(5)*8-30
150 FOR B=1TO 3;TV=47+RND(10);NEXT B;NEXT A
160 NT=0;FC=252
170 CX=25;CY=-39;PRINT "PRESS GO",
175 BOX 63,-39,13,9,3
180 A=RND(10)
190 IF (23)=0GOTO 180
200 (4)=RND(10)-1;(7)=(4)
210 (5)=RND(10)-1
220 IF (4)=(5)GOTO 210
230 (8)=(5)
240 (6)=RND(10)-1
250 IF ((6)-(4))+((6)-(5))GOTO 240
260 CLEAR ;G=0;Q=0+1;FC=7;BC=243
270 CY=40;PRINT "GAME",#3,Q
280 IF U CX=0;CY=40;PRINT "AVERAGE",#3,U,".",#1,F
300 NT=0;CX=-77;PRINT "
310 NT=1;CX=-77;PRINT #2,G+1,">",
320 FOR A=1TO 3
330 X=KP;IF X=31GOTO Z
340 TV=X
350 IF (X<48)-(X>57)GOTO 900
360 (A)=X-48;NEXT A
370 IF (1)+(2)+(3)=0GOTO 750
380 IF ((1)-(2))+((1)-(3))+((2)-(3))GOTO 910
500 G=G+1;P=0
510 FOR A=1TO 3
520 IF (A)=(A+3)P=P+4
530 IF (A)=(A+4)P=P+1
540 IF (A)=(A+5)P=P+1
550 NEXT A
```

```
610 IF P=0PRINT " BAGELS";GOTO Z
620 IF P=1PRINT " PICO";GOTO Z
630 IF P=3PRINT " PICO-PICO";GOTO Z
640 IF P=4PRINT " FERMI";GOTO Z
650 IF P<7PRINT " FERMI-PICO";GOTO Z
660 PRINT " FERMI-FERMI"
670 PRINT "GAME GUESS? ",;A=KP
680 IF A=13GOTO 700
690 PRINT "NO",;NT=30;MU=20;MU=18;MU=18;GOTO Z
700 CX=CX-12;PRINT " ";PRINT ;IF P#12GOTO 740
710 PRINT "YOU GOT IT";W=W+1;BC=155
720 NT=8;A=48;MU=99;MU=53;MU=A;MU=49;MU=A;MU=A
730 MU=50;MU=49;MU=50;MU=51;MU=A;MU=51;MU=A;MU=51;MU=A;
GOTO 790
740 PRINT "SORRY, WRONG NUMBER";D=D+1;BC=66;GOTO 760
750 PRINT ;PRINT ;PRINT "GAVE UP";BC=123
760 PRINT "CORRECT NUMBER WAS ",#1,(4),(5),(6)


770 NT=10;MU=51;MU=45;MU=51;MU=51;MU=45;MU=51;MU=51
780 MU=99;MU=55;MU=50;MU=49;MU=99;MU=54;MU=48;MU=48
790 NT=0
800 C=C+G;IF W=0GOTO 860
810 U=C-W;F=(RM,10)÷W
820 PRINT "STAT. AFTER GAME #",#2,0
830 PRINT " # GAMES WON ....",#2,W
840 PRINT " # GAMES LOST ...",#2,D
850 PRINT " AVE # GUESSES...",#2,U,".",#1,F
860 GOTO 180
900 PRINT " INPUT ERROR";GOTO 920
910 PRINT " DUPLICATE DIGITS",
920 NT=5;FOR A=1TO 10;MU=87;MU=84;NEXT A
930 GOTO Z
```



PROGRAM NAME: CONNECT FOUR

ISSUE: Volume 2 page 35

AUTHOR Larry Camnitz  
2106 Hastings St.  
Orlando, FL 32808



DESCRIPTION-

The object is to get four squares in a row while preventing your opponent from doing the same. Use hand controllers (1) and (2). When four squares are lined up, move the joystick sideways to register the win.

LISTING -

```

10 BC=126
20 FC=0
30 NT=0
900 P=1:L=1;M=0;N=0
950 CLEAR ;GOSUB 3000
960 CLEAR
1000 BOX -4,-36,118,10,2
1002 CX=-70;CY=20;PRINT "GAME",
1004 CX=45;CY=20;PRINT "BLACK"
1005 CX=58;CY=-20;PRINT #0,M
1006 CX=45;CY=-10;PRINT "WHITE"
1007 CX=58;CY=10;PRINT #0,N
1008 CX=-64;CY=10;PRINT #0,L
1010 BOX -4,-3,87,56,1
1020 FOR B=-40TO 32STEP 12
1030 FOR C=20TO -25STEP -9
1040 BOX B,C,11,8,2
1050 NEXT C
1060 NEXT B
1070 X=-40
1080 D=-34;E=-34;F=-34;G=-34;H=-34;I=-34;J=-34
1100 CY=40;IF P=1PRINT "←7→BLACK IS UP";GOTO 1600
1110 CY=40;IF P=2PRINT "      WHITE IS UP";GOTO 1600
1200 IF TR(1)=1GOTO 2000
1210 IF JX(1)=1GOTO 1500
1220 IF JX(1)=-1GOTO 1510
1230 IF JY(2)=1GOTO 3500
1240 GOTO 1200
1400 IF TR(2)=1GOTO 2000
1405 IF &(22)=16GOTO 900
1410 IF JX(2)=1GOTO 1500
1420 IF JX(2)=-1GOTO 1510
1430 IF JY(1)=1GOTO 3600
1440 GOTO 1400
1500 X=X+12;GOTO 1600
1510 X=X-12;GOTO 1600
1600 IF X<-40X=-40
1610 IF X>32X=32
1620 BOX -4,30,100,6,2
1700 BOX X,30,9,6,1
1705 IF P=2BOX X,30,5,4,2
1706 NT=1
1710 MU="Y";MU="Z"
1711 NT=0
1720 IF P=1GOTO 1200
1730 IF P=2GOTO 1400

```

←N→ INDICATES  
"LEAVE  
N  
SPACES"

```

2000 IF X=-40D=D+9;GOTO 2200
2010 IF X=-28E=E+9;GOTO 2300
2020 IF X=-16F=F+9;GOTO 2400
2030 IF X=-4G=G+9;GOTO 2500
2040 IF X=8H=H+9;GOTO 2600
2050 IF X=20I=I+9;GOTO 2700
2060 IF X=32J=J+9;GOTO 2800
2200 BOX X,D,9,6,1
2210 IF P=2BOX X,D,5,4,2;P=1;GOTO 1100
2220 P=2;GOTO 1100
2300 BOX X,E,9,6,1
2310 IF P=2BOX X,E,5,4,2;P=1;GOTO 1100
2320 P=2;GOTO 1100
2400 BOX X,F,9,6,1
2410 IF P=2BOX X,F,5,4,2;P=1;GOTO 1100
2420 P=2;GOTO 1100
2500 BOX X,G,9,6,1
2510 IF P=2BOX X,G,5,4,2;P=1;GOTO 1100
2520 P=2;GOTO 1100
2600 BOX X,H,9,6,1
2610 IF P=2BOX X,H,5,4,2;P=1;GOTO 1100
2620 P=2;GOTO 1100
2700 BOX X,I,9,6,1
2710 IF P=2BOX X,I,5,4,2;P=1;GOTO 1100
2720 P=2;GOTO 1100
2800 BOX X,J,9,6,1
2810 IF P=2BOX X,J,5,4,2;P=1;GOTO 1100
2820 P=2;GOTO 1100
3000 CY=20;PRINT "←6→CONNECT FOUR
3010 PRINT ;INPUT " NUMBER OF GAMES?"K
3020 RETURN
3500 CX=-56;CY=-38;PRINT "WHITE CONNECTS FOUR"
3510 M=M+1;CX=58;CY=-20;PRINT #0,M
3515 L=L+1
3520 IF L-1=KGOTO 4500
3530 IF TR(2)=1GOTO 1000
3540 GOTO 3530
3600 CX=-56;CY=-38;PRINT "BLACK CONNECTS FOUR
3610 N=N+1;CX=58;CY=10;PRINT #0,N
3615 L=L+1
3620 IF L-1=KGOTO 4500
3630 IF TR(1)=1GOTO 1000
3640 GOTO 3630
4500 CY=40;PRINT "←16→GAME OVER "
4510 IF &(22)=16GOTO 900
4520 GOTO 4510

```

Copyright 1980, Robert Fabris



PROGRAM NAME BINGO

ISSUE Volume 2 page 34

AUTHOR Ernie Sams  
248 S. Forest St.  
Bellingham, WA 98225

DESCRIPTION

B	I	N	G	O	B	I	N	G	O
8	23	39	54	62	5	27	45	60	69
10	27	42	50	75	11	19	34	54	67
6	28	H	52	65	14	20	C	58	63
15	16	40	53	63	12	30	44	57	68
11	29	36	60	72	4	24	40	53	64

\*75 UNDER O

The computer plays against you, with one Bingo card each. The computer randomly draws numbers and displays them on the screen. Turn the knob to indicate whether you have the number (yes no) and pull the trigger to register. The computer then checks its card.

LISTING

```

30 CLEAR ;NT=1;FC=179;&(23)=255
40 &(9)=84;&(0)=7;&(1)=7;&(2)=88;&(3)=88
50 BOX -40,9,68,65,1;BOX -40,4,64,49,2
60 BOX 34,9,68,65,1;BOX 34,4,64,49,2
70 CX=-64;CY=36;PRINT "B I N G O",;CX=10;PRINT "B I N G O",
80 FOR L=-60TO -21STEP 13;BOX L,10,1,64,1;BOX L+74,10,1,64,1;NEXT L
90 FOR L=-11TO 19STEP 10;BOX -40,L,68,1,1;BOX 34,L,68,1,1;NEXT L
100 C=0;D=0;X=-69;Y=34
110 FOR L=1TO 125;@(L)=0;NEXT L
120 FOR L=75TO 95STEP 5
130 FOR M=1TO 5
140 N=RND (15)+C
150 IF (@(L+1)=N)+(@(L+2)=N)+(@(L+3)=N)+(@(L+4)=N)+(@(L+5)=N)GOTO 140
160 @(L+M)=N
170 CX=X;CY=Y+(-M*10);PRINT #2,N,
180 N=RND (15)+C
190 IF (@(L+26)=N)+(@(L+27)=N)+(@(L+28)=N)+(@(L+29)=N)+(@(L+30)=N)GOTO 180
195 @(L+25+M)=N
200 CX=X+74;CY=Y+(-M*10);PRINT #2,N,
210 NEXT M
220 C=C+15;X=X+13
230 NEXT L
232 FOR L=1TO 3;CY=4;CX=-40;PRINT "H",;CX=34;PRINT "C",
233 BOX -40,4,12,9,L;BOX 34,4,12,9,L;NEXT L
234 @(88)=-1;@(113)=-1
240 FOR L=1TO 75
250 N=RND (75);IF @(N)<0GOTO 250
260 @(N)=-1
262 CX=-72;CY=-32
264 IF N<16PRINT "# ",#2,N," UNDER B",
266 IF N>15IF N<31PRINT "# ",#2,N," UNDER I",
268 IF N>30IF N<46PRINT "# ",#2,N," UNDER N",
270 IF N>45IF N<61PRINT "# ",#2,N," UNDER G",
272 IF N>60PRINT "# ",#2,N," UNDER O",
278 BOX 34,-32,70,10,2
280 IF KN(1)<0PRINT " NO? ",
290 IF KN(1)>=0PRINT " YES?",
300 IF TR(1)=0GOTO 262
310 IF KN(1)<0GOTO 340
320 A=0;U=0;GOSUB 400
330 GOSUB 600
340 A=25;U=74;GOSUB 400
350 GOSUB 600
360 NEXT L
400 FOR S=76+ATO 100+A;IF @(S)=NGOTO 500
410 NEXT S

```

## BINGO (continued)

```


430 IF A=25RETURN
440 PRINT " NOPE",;RETURN
500 X=-66
510 FOR S=75+ATO 95+ASTEP 5
512 Y=24
520 FOR T=1TO 5
530 IF @(S+T)=N&(21)=255;BOX X+U,Y,12,9,3;@(S+T)=-1;&(21)=0;RETURN
540 Y=Y-10;NEXT T
550 X=X+13;NEXT S
560 RETURN
600 FOR S=76+ATO 96+ASTEP 5
610 IF @(S)=-1IF @(S+1)=-1IF @(S+2)=-1IF @(S+3)=-1IF @(S+4)=-1GOTO 800
620 NEXT S
630 FOR S=76+ATO 80+A
640 IF @(S)=-1IF @(S+5)=-1IF @(S+10)=-1IF @(S+15)=-1IF @(S+20)=-1GOTO 800
650 NEXT S
660 FOR S=76+ATO 100+ASTEP 6
670 IF @(S)≠-1GOTO 700
680 NEXT S
690 GOTO 800
700 FOR S=80+ATO 96+ASTEP 4
710 IF @(S)≠-1RETURN
720 NEXT S
800 CY=-40
810 FOR L=0TO 9
820 IF A=0CX=-72;PRINT "BINGO--YOU WIN",
830 IF A=25CX=8;PRINT "BINGO--I WIN",
840 NEXT L
  
```



**PROGRAM NAME HORSERACE**

**ISSUE** Volume 3 page 26

**AUTHOR** Paul Slezak  
1104 S. Neltnor  
West Chicago, IL 60185



**DESCRIPTION**

Eight horses are available to bet upon, using the keypad to enter your horse (B) and the value of your bet (C). The computer then randomly moves the horses across the screen and calculates the winnings.

**LISTING**

```

5 NT=0;CLEAR ;FC=155;BC=160;N=0;FOR Z=1TO 4;@(Z+20)=500;NEXT Z;&(9)=180
10 CY=5;PRINT "WELCOME TO ARLINGDUNG PARK";INPUT "# OF PLAYERS? (1-4)*A;IF A>4
GOTO 10
15 GOSUB 290
20 CLEAR ;CX=-20;PRINT "RACE #";PRINT #1,N;T=0;S=0;GOSUB 600
30 FOR Z=1TO 8;PRINT #2,Z,;@(Z+8)=RND (5)+RND (3)+1;CX=38;PRINT #1,@(Z+B),;CX=
45;PRINT ":1",;CX=-50;GOTO 30+Z
31 PRINT "ALBADEXTER";NEXT Z
32 PRINT "JO L IN";NEXT Z
33 PRINT "C BICUSPID";NEXT Z
34 PRINT "SEATTLE SLEWED";NEXT Z
35 PRINT "WOMAN O' WAR";NEXT Z
36 PRINT "DUSTY PAUL";NEXT Z
37 PRINT "DIABLO";NEXT Z
38 PRINT "SECRETARYLESS";NEXT Z
40 FOR Z=1TO A;IF @(Z+20)=0@ (Z+16)=0;NEXT Z;GOTO 100
50 CY=-32;PRINT "PLAYER #";PRINT #1,Z,;CX=-20;INPUT B;CY=-32;CX=10;INPUT C
60 IF B>0IF B<10GOTO 80
70 GOTO 50
80 IF C>@(Z+20)CY=-32;PRINT "SORRY,I'M NO LOAN SHARK";GOSUB 610;CY=-32;FOR D=1
TO 23;PRINT " ",;NEXT D;CX=-78;GOTO 50
90 @(Z+16)=B:@ (Z+20)=@(Z+20)-C:@ (Z+24)=C;NEXT Z
100 FC=160;NT=5;PRINT "146 641 641 641000000";CLEAR ;NT=0;FC=155;CY=35
110 FOR Z=1TO 10;CX=70;GOTO 110+Z
111 PRINT "X";NEXT Z
112 PRINT "X";NEXT Z
113 PRINT "F";NEXT Z
114 PRINT "I";NEXT Z
115 PRINT "N";NEXT Z
116 PRINT "I";NEXT Z
117 PRINT "S";NEXT Z
118 PRINT "H";NEXT Z
119 PRINT "X";NEXT Z
120 PRINT "X";NEXT Z
130 CY=25;FOR Z=1TO 8;PRINT #2,Z,;CX=-60;PRINT " ";NEXT Z;GOSUB 610
140 CY=25;FOR Z=1TO 8;CX=-60;PRINT " ";NEXT Z
150 NT=1;&(20)=50;&(21)=205;FOR Z=1TO 35;MU="U";NEXT Z;NT=0;&(20)=0;&(21)=0;CY=
25;FOR Z=1TO 8;PRINT " ";NEXT Z
155 CY=0;PRINT " ..AND THEY'RE OFF!!!";GOSUB 610;CY=0;PRINT " ←————— 19 SPACES —————→ "
160 GOSUB 610
170 CY=25;FOR Z=1TO 8;@(Z)=@(Z)+((RND (4)+B+RND (5))-(RND (@ (Z+8)))));CX=@ (Z);PR
INT #1,Z;IF @(Z)>65S=1
175 NEXT Z;GOSUB 610
180 IF S=1GOTO 200
190 CY=25;GOSUB 620;GOTO 160

```

Copyright 1980, Robert Fabris



# ARCADIAN

## HORSERACE (continued)

```
200 G=0;FOR Z=1TO 8;IF G>@(Z)NEXT Z;GOTO 220
210 X=Z;G=@(Z);M=@(Z+8);NEXT Z
220 B=0;FOR Z=1TO 8;IF @(Z)>65B=B+1
230 NEXT Z;IF B>1GOSUB 500
240 CLEAR ;CY=0;PRINT "THE WINNER IS #",;PRINT #2,X;GOSUB 610
250 FOR Z=1TO A;IF @(Z+16)=X@(Z+20)=@(Z+24)M+@(Z+20)
260 NEXT Z
290 CLEAR ;CX=-30;PRINT "YOU HAVE"
300 FOR Z=1TO A;PRINT "PLAYER #",;PRINT #1,Z;PRINT #2," $",;PRINT #2,@(Z+20);
NEXT Z
310 FOR Z=1TO 3000;NEXT Z
320 IF N>8GOTO 400
330 N=N+1;IF N>0GOTO 20
340 RETURN
400 CLEAR ;FC=79;BC=82;CY=16;CX=-35;PRINT "RACES OVER!!";PRINT ;PRINT
410 PRINT "IF YOU WANT 9 MORE RACES PRESS 1";A=KP;IF A#49STOP
420 GOTO 5
500 FOR Z=1TO 6;CLEAR ;FC=212;BC=209;CY=0;CX=-35;PRINT "PHOTO FINISH";NEXT Z;
BC =160;FC=155
510 RETURN
600 FOR Z=1TO 8;@(Z)=-78;NEXT Z;RETURN
610 FOR F=1TO 300;NEXT F;RETURN
620 FOR D=1TO 8;CX=@(D);PRINT " ";NEXT D;RETURN
```




PROGRAM NAME: FIFTEEN

ISSUE: Volume 2 page 64

AUTHOR Bob Wiseman  
6594 Swartout Rd.  
Algonac, MI 48001

DESCRIPTION-



Based on the familiar 15 puzzle, this version uses letters. They are randomly placed by the computer, and you use the joystick to move the letters into the empty space, in order to place them all in alphabetical order.

LISTING-

```

10 FC=11;U=400;CLEAR
12 NT=1
15 BOX -2,0,75,75,1
16 BOX -2,0,73,73,3
20 E=0;A=25
30 FOR Y=1TO 24
31 @(Y)=Y+64
35 GOSUB 400;NEXT Y
40 @(25)=32;GOSUB 400
50 IF TR(1)=1U=1
60 IF E=0B=RND(4);GOTO 100
70 I=JX(1);J=JY(1)
80 IF I=0IF J=0GOTO 70
90 IF IIF JGOTO 70
91 B=0
92 IF J B=J+2
94 IF I B=3-I
100 X=A-A:5*5
110 IF B=2IF X=1GOTO 300
120 IF B=4IF X=0GOTO 300
130 IF B=3IF A>20GOTO 300
140 IF B=1IF A<6GOTO 300
150 IF B=1D=A-5
160 IF B=2D=A-1
170 IF B=3D=A+5
180 IF B=4D=A+1
190 X=@(D);@(D)=32;@(A)=X
192 IF E=1GOTO 200
193 U=U-1;A=D
194 IF U<200FOR Y=1TO 25;GOSUB 400;NEXT Y;E=1;GOTO 50
196 GOTO 50
200 Y=D;GOSUB 400
210 Y=A;GOSUB 400
220 A=D
221 IF U=0GOTO 50
222 U=0
225 Y=0
230 FOR X=1TO 25
235 IF X=AGOTO 250
240 IF @(X)<YGOTO 50
245 Y=@(X)
250 NEXT X
260 FOR X=1TO 25
270 MU=X;NEXT X
280 GOTO 10
300 IF E=0GOTO 50
302 FOR X=1TO 5
304 MU=X;NEXT X
310 GOTO 50
400 MU=@(Y)
405 Z=Y-1
410 CX=-30+(Z-Z+5*5)*13
420 CY=30-Z+5*14
430 TV=@(Y)
440 RETURN

```




PROGRAM NAME LOGO

AUTHOR Guy McLimore  
 ABC Hobbycraft  
 2155 E. Morgan Ave.  
 Evansville, IN 47711

DESCRIPTION:

ISSUE Vol 2 Page 3



This is a graphics program that projects a random star field on the screen, then prints a word. A tail erupts from the end of the word, which in turn becomes surrounded by a rocket. We use this as the logotype for the Newsletter.

LISTING:

```

10 CLEAR
20 &(9)=37
30 BC=250;FC=7
40 FOR A=1 TO 150
50 H=RND (160)-80;U=RND (88)-44
60 BOX H,U,1,1,RND (2)
70 NEXT A
80 &(9)=85;&(1)=250
90 &(2)=108;&(11)=250
100 CX=-44;CY=0
110 PRINT "ARCADIAN
120 LINE 0,0,4
130 FOR Q=0 TO 35 STEP 2
140 LINE 75,Q,3;LINE 0,0,4
150 LINE 75,-Q,3;LINE 0,0,4
160 NEXT Q
170 L=7;FOR Z=48 TO 51
180 BOX -Z,0,1,L,1
190 L=L-2
200 NEXT Z
210 BOX -22,0,50,9,3
220 FOR Q=0 TO 35
230 LINE 75,Q,3;LINE 0,0,4
240 LINE 75,-Q,3;LINE 0,0,4
250 NEXT Q
260 GOTO 220
>
  
```



PROGRAM NAME MICROTREK

ISSUE Volume 1, page 89; V2p4

AUTHOR Bill Andrus  
7034 Thomas Dr.  
North Highlands, CA 95660

1	2	3	4	5	6	7	8	
1	.	.	.	.	*	.	.	1
2	.	*	.	.	.	.	K	2
3	.	.	.	*	.	.	.	3
4	.	.	.	.	.	.	.	4
5	.	.	E	.	*	.	.	5
6	.	*	.	.	.	.	K	6
7	.	.	.	.	.	*	.	7
8	.	.	K	.	.	.	.	8
1	2	3	4	5	6	7	8	

COMMAND:

DESCRIPTION

A small version of Star Trek. The Space is an 8x8 quadrant of 64 Sectors. You can move to any legal, unoccupied sector. If you are adjacent to a Star Base, you are docked and restoring energy. The Commands are:

- 1: Move to Sector (row,column) If either command is zero, the command is cancelled
- 2: Move to a new Quadrant, extending your search for Klingons and Star Bases.
- 3: fire on Sector (row,column) - watch your energy
- 4: Sensor report - \* are Stars K is Klingons E is Enterprise B is Star Base
- 5: Status report - gives display

LISTING

```

30 CLEAR ;NT=0
40 D=RND (40);IF D<30GOTO 40
50 W=RND (35);IF W<10GOTO 50
60 D=(D+W)/20;L=10000
70 PRINT " *MICROTREK*
100 A=0;K=0;S=10;T=10
110 FOR I=1TO 64;@(I)=0
120 X=RND (12);IF X>2 @(I)=1
130 IF X=3 IF K<W @(I)=2;K=K+1
140 NEXT I
150 X=RND (64);IF @(X)>1 GOTO 150
160 @(X)=3;GOSUB 850;B=RND (17);IF B>5 GOTO 200
170 X=RND (64);IF @(X)>1GOTO 170
180 @(X)=4;S=(X-1)/B;T=X-S*B
200 CLEAR ;C=E-S;IF C<0C=-C
210 G=F-T;IF G<0 G=-G
220 Q=0;IF C<2 IF G<2 Q=1
230 D=D-1;IF D=0 GOTO 970
240 IF (K=0)+(Q=1) GOTO 270
250 GOSUB 860;PRINT #4,H,"UNIT HIT FROM"
260 PRINT "KLINGONS!"
270 IF Q=1 L=10000
280 INPUT "COMMAND: "A
290 IF A=1 GOTO 350
300 IF A=2 GOTO 400
310 IF A=3 GOTO 450
320 IF A=4 GOTO 550
330 IF A=5 GOTO 650
340 GOTO 280
350 GOSUB 920;X=(Y-1)*B+Z;IF @(X)#1 PRINT "SECTOR OCCUPIED";GOTO 350
360 U=1;GOSUB 870;@(X)=3;@(E*B+F)=1;GOSUB 850;GOTO 200
400 G=RND (250)+300;U=1;GOSUB 900;GOTO 100
450 GOSUB 920;U=2;GOSUB 870;X=(Y-1)*B+Z;U=@(X);IF U<2 GOTO 230
460 IF U=3 PRINT "YOU DESTROYED YOURSELF!";GOTO 990
470 IF U=4 PRINT "STARBASE DESTROYED!";S=10;T=10;GOTO 200
480 IF R>500 IF (RND (R+500)+6)>10 PRINT "YOU MISSED!";GOTO 230
490 @(X)=1;K=K-1;W=W-1;IF W>0 GOTO 230
500 PRINT "MISSION ACCOMPLISHED!";GOTO 990
550 CLEAR ;GOSUB 630
560 FOR I=1TO 8;PRINT #1,I;FOR J=1TO 8
570 X=@((I-1)*B+J);IF X=0 PRINT " * ",
580 IF X=1 PRINT " . ",
590 IF X=2 PRINT " K ",
600 IF X=3 PRINT " E ",
610 IF X=4 PRINT " B ",
620 NEXT J;PRINT #1,I;NEXT I;GOSUB 630;GOTO 280
650 CLEAR ;PRINT " STATUS REPORT"
660 PRINT "SECTOR:",#17,E+1,"",#1,F
670 PRINT "STARDATE:",#17,D
680 PRINT "ENERGY:",#19,L;PRINT "KLINGONS:",#17,W
690 PRINT "CONDITION: ",;GOSUB 700;GOTO 280
700 IF Q=1PRINT "DOCKED";FC=0;RETURN
710 IF K>0PRINT " RED";FC=90;RETURN
720 IF L=2000PRINT " GREEN";FC=172;RETURN
730 PRINT "YELLOW";FC=133;RETURN
850 E=(X-1)/B;F=X-E*B;RETURN
860 H=(RND (50)+200)*K;G=H;U=1;GOTO 900
870 R=((Y-E)*(Y-E))+((Z-F)*(Z-F))*100
880 G=R/10;IF G=0 RETURN
890 J=G;G=(R+G)/2;IF G<J GOTO 890
900 L=L-U*G;IF L>0 RETURN
910 PRINT "OUT OF ENERGY!";GOTO 990
920 INPUT "SECTOR ROW: "Y;IF Y=0 GOTO 280
930 IF (Y<1)+(Y>8) GOTO 920
940 INPUT "SECTOR COLUMN: "Z;IF Z=0 GOTO 280
950 IF (Z<1)+(Z>8) GOTO 940
960 RETURN
970 PRINT "OUT OF TIME!";GOTO 990
990 PRINT "GAME OVER."

```

PROGRAM NAME NICHOMACHUS

AUTHOR Hank Chiuppi  
275 St. Mary's  
Buffalo Grove, IL 60090

DESCRIPTION:

Volume 2 page 72

HELLO! I AM THE BALLY  
COMPUTER.  
HOW MANY LETTERS IN  
YOUR NAME?

The computer attempts to guess a number you have chosen by asking three questions. The player has to do some arithmetic to supply the answers.

LISTING:

```

10 CLEAR
20 PRINT "HELLO! I AM THE BALLY      COMPUTER.
30 INPUT " HOW MANY LETTERS IN YOUR  NAME?"L
40 IF L<5PRINT "MY, YOU HAVE A VERY SHORT NAME!"
50 IF L>15PRINT "WOW! YOU HAVE A VERY LONG NAME!"
60 PRINT " BY THE WAY, WHAT IS YOUR  NAME?"
70 FOR A=1TO L;B=KP;@(A)=B
80 NEXT A
90 PRINT "HELLO!";GOSUB 1000
100 GOSUB 2000
110 PRINT " --OOPS! SORRY";PRINT "ABOUT THAT.
120 GOSUB 2000
130 PRINT "HELLO!";GOSUB 3000
140 GOSUB 2000
150 PRINT " --THAT'S BETTER.
160 INPUT "WOULD YOU LIKE TO PLAY A  GAME? (1=YES  2=NO)"G
170 IF G=1GOTO 200
180 PRINT "IT WAS NICE MEETING YOU";GOSUB 1000
190 STOP
200 PRINT "OK ";GOSUB 3000
210 PRINT " --PICK A NUMBER";PRINT "BETWEEN 7 AND 100";PRINT "DON'T TELL ME
    WHAT IT IS.
220 GOSUB 2000
230 INPUT "YOUR # ÷3 HAS A REMAINDER  OF?"Q
240 INPUT "YOUR # ÷5 HAS A REMAINDER  OF?"R
250 INPUT "YOUR # ÷7 HAS A REMAINDER  OF?"S
255 D=0
260 D=(70×Q)+(21×R)+(15×S)
270 IF D<=105GOTO 300
280 D=D-105
290 GOTO 270
300 PRINT "YOUR # IS ",D;PRINT "RIGHT?
310 INPUT "(1=YES  2=NO)"X
320 IF X=1GOTO 340
330 PRINT "I THINK YOUR ARITHMETIC   IS IN ERROR!!
340 PRINT "WANT TO TRY AGAIN?";GOSUB 3000
350 INPUT "(1=YES  2=NO)"Y
360 IF Y=1GOTO 200
370 GOTO 180
1000 FOR C=LTO 1STEP -1
1010 TV=@(C)
1020 NEXT C
1030 RETURN
2000 FOR T=1TO 2000
2010 NEXT T
2020 RETURN
3000 FOR C=1TO L;TV=@(C)
3010 NEXT C
3020 RETURN

```

Copyright 1980, Robert Fabris

PROGRAM NAME REVERSE

AUTHOR Brett Bilbrey & Mike Toth  
14430 Barclay  
Dearborn, MI 48126

DESCRIPTION:

Volume 1 page 38,46

74

6159328

REVERSE DIGITS 1 THRU

Nine numbers are randomly placed in line. The object is to arrange them in numerical order (smallest to the left). The knob of hand controller (1) is used to indicate how many numbers will be moved, and the trigger used to make the move.

LISTING:

```

9 NT=9
10 N=9;CLEAR
20 @(1)=RND (9)
30 FOR K=2TO N
40 @(K)=RND (9)
50 FOR J=1TO K-1
60 IF @(K)=@(J)GOTO 40
70 NEXT J;NEXT K
80 CY=25
90 PRINT "THE LIST IS"
100 T=0
110 GOSUB 280
120 CY=-20;CX=0;Q=0;NT=0
130 O=KN(1)÷32+6;IF TR(1)=1GOTO 160
135 CY=-33
140 IF O#Q CX=-77;PRINT "REVERSE DIGITS 1 THRU ",#1,0;Q=0
150 GOTO 130
160 T=T+1
170 FOR K=1TO Q÷2
180 Z=@(K)
190 @(K)=@(Q-K+1)
200 @(Q-K+1)=Z
210 NEXT K
220 GOSUB 280
230 FOR K=1TO 9;IF @(K)#KGOTO 120
240 NEXT K
250 CX=-70
255 CLEAR
260 CY=0;PRINT " YOU WON IN ",;TV=T÷10+48;TV=T-T÷10*10+48;PRINT " MOVES"
265 PRINT " PULL TR(1) TO ←—14 sp.—→ PLAY AGAIN"
270 IF TR(1)#1GOTO 270
275 GOTO 10
280 CX=-5;CY=0
290 FOR B=1TO N;TV=48+@(B);TV=32;NEXT B
300 RETURN


```

PROGRAM NAME:     SPIRALS II    

AUTHOR      Matt Giwer  
             3922 Millcreek Dr.  
             Annandale, VA 22003

DESCRIPTION:

Volume 2 page 95



This is an artistic exercise that draws patterns on the screen, using a diamond motif as generated by a line following a spiral path around the center.

LISTING:

```

10 CLEAR
100 A=3
200 B=RND (2)
210 C=RND (2)
800 X=B;Y=0
850 FOR D=1TO 100
900 X=X+B
905 Y=Y+C
907 IF X>75GOTO 100
908 IF Y>45GOTO 100
910 GOSUB 1050
950 NEXT D
990 GOTO 10
1050 .M
1060 LINE X,0,A
1070 LINE 0,-Y,A
1090 LINE -X,0,A
1110 LINE 0,Y,A
1200 RETURN
>

```



# ARCADIAN

ARCADIAN SAMPLER PROGRAMS, as included herein, are also available on a single tape. It can be purchased in either the ARCADE Plus format (where the cassette interface is on the Basic cartridge), or in the Bally Basic format (where there is a separate cassette interface). Either cartridge is \$7.50

OTHER DOCUMENTATION available includes the disassembled listings of the Basic languages. Advanced hackers will find these of value in developing more complex programs. The Bally Basic listing is \$6.50, while the new Arcade Plus listing is \$7.50.

TAPED PROGRAMS are also available of other programs that have appeared in the ARCADIAN. We have tapes of the "BEST OF ARCADIAN" for 1979, 1980, and 1981. Inquire as to contents and prices.

ARCADIAN  
3626 Morrie Drive  
San Jose, CA 95127-9990

408-272-1060  
The SOURCE TCD959

# ARCADIAN



a = →

b = ×

c = ÷

## NOTES

These NOTES are provided to introduce the new ARCADE owner to techniques and effects that can be generated by the Tiny Basic cartridge. Considerably more detail can be found in the ARCADIAN articles referred to in the text. In addition, a listing of available publications that can assist the more advanced programmer is included.

The ARCADIAN articles were written as descriptions of the original "Bally Basic". While this cartridge has been superseded by the current "AstroVision Basic", the explanatory material remains applicable. As new material is discovered and developed by experimenters, it will be documented in the ARCADIAN.

Your inputs are solicited, as production of the ARCADIAN is almost totally based on subscriber inputs.

**SCREEN CHARACTERS** The Arcade divides the tv screen into 16320 individual dots, called pixels. This is disposed at 102 high by 160 wide. The lettering of the Arcade is 5 pixels wide-plus one, and 7 pixels high-plus one. (The 'one' is to prevent adjacent letters from running together.) A full explanation of controlling location is defined in Vol. 1, page 50, while methods to drive the lettering size upwards start on page 45.

Home-brewed characters can be generated by means of a method defined in Vol. 3, page 83. In this program, a 16x10 matrix can be selectively filled in to create a character. A further expansion of this technique is described in Vol. 3, page 128, wherein the computer is made to do most of the work.

**MEMORY DUMPS** The Basic can be asked to identify what is located within its memory registers in binary or hexadecimal notations by some simple programs. (binary, Vol 1, p.43; hex, page 44, for example)

**SOUND GENERATION** lengthy explanation of the operation of the sound systems is found in Vol. 1, pages 62-66 and 70-73

**MUSIC GENERATION** (using the available three-tone system) is a subset of the sound generation system, and is covered by articles in Vol. 2, page 62, and Vol 4, page 10

**BOOLEAN MATH** is utilized in some functions and commands, and is discussed in Vol.1, pages 41,44, 52,55, amongst others.

&(10) is a command that gives you control over an apparent "curtain" on the screen, behind which you can place words or characters until you are ready to display them. Try

```
FOR A = 0 TO 180; &(10)=A; NEXT A
```

&(9) is a command that gives you control over the ability to divide the screen vertically. Try the same program as above, only replace &(10)=A with &(9)=A. Then try adding this command at the beginning: &(0)=172; &(11)=126; &(2)=82 for a colorful surprise. These are described in Vol. 1 pages 15, 40.

MACHINE CODE programs are possible in the Arcade:-

%(n) activates the PEEK/POKE relationship. In order to perform a PEEK function, where one observes what is stored in a memory location, one commands A=%(nnnn) then a PRINT A will result in the value located in location nnnn

To place A into location nnnn, one commands %(nnnn)=A.

CALL is a direct command to intercept a program within the system ROM. Try CALL 3177 or CALL 4910.

Putting all of this together, and knowing how the insides of the Arcade works, enables one to write programs in machine code directly from the Keypad. The first article discussing this is in Vol. 1, page 25. There are a number of subsequent articles touching on various parts of the problem.

**PUBLICATIONS** The following papers are available. At the moment, these document the Arcade with the Bally Basic installed. Exactly how much of these are applicable for the AstroVision Basic is anyone's guess. They are included for your information, and as we develop our understanding of the AstroVision Basic, we shall update, revise, or reprint the documents.

**EXECUTIVE SOFTWARE** - a listing of software that does something in the Arcade, and how to execute it. Part of the document lists the ROM subroutines that are executed with an RST 3BH instruction, while the second part contains the listings of the on-board ROM from 0000-1FFF. 27 pages.

**ASTROVISION BASIC** a completely disassembled listing of the Basic cartridge on 70 pages, with comments. (We have a similar listing of the Bally Basic.)

**BALCHEK** A program was developed by Bally software engineers which "looked at" the operation of the printed circuit board and then determined if any problems existed, all the while doing a burn-in operation. It would then identify the errant problem area. The listing and its instructions in over 60 pages. This program is also available on a chip, and in a complete, ready to operate tool.

**MANUAL OF HARDWARE AND SOFTWARE** A large document made up by the Bally software design engineers explaining a number of routines, machine operations, and details of the inner workings, including some specifications of the three custom chips. The second half of the document contains disassembled listings. 300 pages.

ARCADIAN  
3626 Morrie Drive  
San Jose, CA 95127-9990